Tema 4
Sucesiones numéricas

Objetivos

1. Definir sucesiones con wxMaxima.
2. Calcular elementos de una sucesión.
3. Realizar operaciones con sucesiones.
4. Interpretar la definición de límite de una sucesión.
5. Calcular el límite de una sucesión.
6. Aplicación práctica de los criterios de convergencia y de cálculos de límites.
7. Aplicación práctica de las sucesiones indeterminadas o indeterminaciones.
8. Aplicación práctica del concepto de sucesiones equivalentes y sus aplicaciones al cálculo de límites.

Contenidos

04-1. Definición de sucesiones. Operaciones con sucesiones.
04-2. Cálculo del límite de una sucesión.
04-3. Formas indeterminadas o indeterminaciones
04-4. Criterios de convergencia y cálculo de límites de sucesiones numéricas.
04-5. Equivalencia de sucesiones.

Referencias

AEM11 ALANINOS PRATS, J; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P. (2011)
Cálculo con wxMaxima.

APJ11 ALANINOS PRATS, J; APARICIO DEL PRADO, C; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P.; VILLENA MUÑOZ, A.R. (2011)
Prácticas de ordenador con wxMaxima.

BR09 BRUZÓN GALLEGO, M. DE LOS SANTOS; RAMÍREZ LABRADOR, JOSÉ (2009)
Modelos matemáticos con Maxima

BU07 DE BURGOS, JUAN (2007)
Cálculo Infinitesimal de una variable (segunda edición).

ES08 ESTELA CARBONELL, M. ROSA; SAÀ SEOANE, JOEL (2008)
Cálculo con soporte interactivo en moodle.

ES02 ESTEP, DONALD (2002)
Practical Analysis in one variable

JB01 JARAUTA BRAGULAT, EUSEBI (2001)
Anàlisi Matemàtica d'una variable. Fonaments i aplicacions.

RR05 REDONDO NEBLE, M. VICTORIA; RODRÍGUEZ GALVÁN, J. RAFAEL (2005)
Introducción a Maxima

RR08b RODRÍGUEZ RIOTORTO, MARIO (2008)
Curso intensivo i-MATH de software libre orientado a Ciencias e Ingeniería

SP95 SPIVAK, MICHAEL (1995)
Calculus (Càlcul Infinitesimal).

VR09 VALLEJO RODRÍGUEZ, JOSÉ ANTONIO (2009)
Cálculo diferencial con Maxima

04-1.- Definición de sucesiones. Operaciones con sucesiones

Los cálculos de este apartado se desarrollan en el fichero **Tema_04-1.wxm**.

La manera habitual de expresar una sucesión es mediante una función de variable discreta, es decir, una función definida como una aplicación del conjunto de los números naturales en un subconjunto de los números reales, tal que a cada número natural $n \in \mathbb{N}$ se le asigna un único número real $x_n \in \mathbb{R}$. La notación habitual para designar una sucesión es $(x_n)_{n\in\mathbb{N}}$ o simplemente (x_n). El número real x_n se llama término general o término *n*-ésimo de la sucesión.

La sintaxis para definir una sucesión con wxMaxima es sencilla; hay que asignar una denominación al término general de la sucesión y escribir a continuación la expresión de este término en función del número natural *n*; por ejemplo, la sucesión $\left(\frac{2n-1}{3n+2}\right)_{n\in\mathbb{N}}$ se define mediante:

```
(%i1) s01[n]:=(2*n-1)/(3*n+2);
```

$$\text{(\%o1) } s01_n := \frac{2\,n-1}{3\,n+2}$$

Se pueden obtener algunos términos de la sucesión, explicitando el número natural antiimagen del término de la sucesión; por ejemplo, los cinco primeros términos de la sucesión anterior son:

```
(%i2) s01[1];    s01[10];     s01[20];
```

$$\text{(\%o2) } \frac{1}{5}$$

$$\text{(\%o3) } \frac{19}{32}$$

$$\text{(\%o4) } \frac{39}{62}$$

Si se quiere obtener una lista con unos cuantos términos de la sucesión, esto se puede obtener mediante la instrucción “makelist”, que se puede concretar con la instrución “numer” para indicar que se quieren los términos en representación decimal; así, por ejemplo:

```
(%i5) makelist(s01[n], n, 1, 10);
```

$$\text{(\%o5) } \left[\frac{1}{5},\frac{3}{8},\frac{5}{11},\frac{1}{2},\frac{9}{17},\frac{11}{20},\frac{13}{23},\frac{15}{26},\frac{17}{29},\frac{19}{32}\right]$$

```
(%i6) makelist(s01[n], n, 1, 5),numer;
(%o6) [0.2,0.375,0.45454545454545,0.5,0.52941176470588]
```

Otros ejemplos de sucesiones y su definición con wxMaxima:

(%i7) s02[n]:=sin(1/n);

(%o7) $s02_n := \sin\left(\frac{1}{n}\right)$

(%i8) makelist(s02[n], n, 1,5);

(%o8) $[\sin(1), \sin\left(\frac{1}{2}\right), \sin\left(\frac{1}{3}\right), \sin\left(\frac{1}{4}\right), \sin\left(\frac{1}{5}\right)]$

(%i9) s03[n]:=sin(n);

(%o9) $s03_n := \sin(n)$

(%i10) makelist(s03[n], n, 1, 5);

(%o10) $[\sin(1), \sin(2), \sin(3), \sin(4), \sin(5)]$

(%i11) s04[n]:=(-1)^n;

(%o11) $s04_n := (-1)^n$

(%i12) makelist(s04[n], n, 1, 5);

(%o12) $[-1, 1, -1, 1, -1]$

(%i13) s05[n]:=(-1)^n/n;

(%o13) $s05_n := \frac{(-1)^n}{n}$

(%i14) makelist(s05[n], n, 1, 5);

(%o14) $[-1, \frac{1}{2}, -\frac{1}{3}, \frac{1}{4}, -\frac{1}{5}]$

(%i15) s06[n]:=n^2;

(%o15) $s06_n := n^2$

(%i16) makelist(s06[n], n, 1, 5);

(%o16) $[1, 4, 9, 16, 25]$

(%i17) s07[n]:=-n^2+3;

(%o17) $s07_n := -n^2 + 3$

(%i18) makelist(s07[n], n, 1, 5);

(%o18) $[2, -1, -6, -13, -22]$

(%i19) s08[n]:=1+(-1)^n/n;

(%o19) $s08_n := 1 + \frac{(-1)^n}{n}$

(%i20) makelist(s08[n], n, 1, 5);

(%o20) $[0, \frac{3}{2}, \frac{2}{3}, \frac{5}{4}, \frac{4}{5}]$

Otra manera de definir o expresar una sucesión es por recurrencia, que consiste en dar una expresión del valor del término general en función de uno o más términos anteriores i concretar los valores iniciales que hacen falta para tener correctamente definida la sucesión. Una sucesión definida así se llama recurrente o definida por recurrencia.

Per ejemplo, la sucesión definida por recurrencia mediante:

$$\begin{cases} x_1 = 3 \\ x_n = \sqrt{2 + x_{n-1}}, \ \forall n \geq 2 \end{cases}$$

se define con wxMaxima de la forma siguiente:

(%i21) s10[1]:3$ s10[n]:=sqrt(2+s10[n-1]);

(%o22) $s10_n := \sqrt{2 + s10_{n-1}}$

(%i23) makelist(s10[n], n, 1, 5);

(%o23) $[3, \sqrt{5}, \sqrt{\sqrt{5}+2}, \sqrt{\sqrt{\sqrt{5}+2}+2}, \sqrt{\sqrt{\sqrt{\sqrt{5}+2}+2}+2}]$

Otro ejemplo de sucesión recurrente: la sucesión definida por:

$$\begin{cases} x_1 = 3 \\ x_n = \dfrac{1}{2}\left(x_{n-1} + \dfrac{3}{x_{n-1}}\right), \ \forall n \geq 2 \end{cases}$$

En wxMaxima se define mediante:

(%i24) s12[1]:3$ s12[n]:=(1/2)*(s12[n-1]+3/s12[n-1]);

(%o25) $s12_n := \frac{1}{2}\left(s12_{n-1} + \frac{3}{s12_{n-1}}\right)$

(%i26) makelist(s12[n], n, 1, 5);

(%o26) $[3, 2, \frac{7}{4}, \frac{97}{56}, \frac{18817}{10864}]$

Y ahora un clásico de este tipo de sucesiones: la llamada sucesión de Fibonacci, definida mediante:

$$\begin{cases} x_1 = 1, \quad x_2 = 1 \\ x_n = x_{n-2} + x_{n-1}, \ \forall n \geq 3 \end{cases}$$

```
(%i27) FIB[1]:1;          FIB[2]:1;          FIB[n]:=FIB[n-1]+FIB[n-2];
(%o27) 1
(%o28) 1
(%o29) FIB_n := FIB_{n-1} + FIB_{n-2}
(%i30) makelist(FIB[n], n, 1, 10);
(%o30) [1,1,2,3,5,8,13,21,34,55]
```

Se pueden llevar a cabo operaciones algebraicas con sucesiones, mediante operaciones con sus términos. Estas operaciones son:

Suma / diferencia:	$(x_n)\pm(y_n)=(x_n\pm y_n)$
Producto:	$(x_n)\cdot(y_n)=(x_n y_n)$
Producto por escalar:	$\lambda\cdot(x_n)=(\lambda x_n),\ \lambda\in\mathbb{R}$
Cociente:	$(x_n)/(y_n)=(x_n/y_n),\ y_n\neq 0,\forall n\in\mathbb{N}$

La sintaxis para hacer estas operaciones con wxMaxima es sencilla: se define una nueva sucesión especificando las operaciones y las sucesiones que operan. A continuación se ponen ejemplos de suma, diferencia, combinación de estas operaciones, producto y cociente:

```
(%i31) s20[n]:=s01[n]+s03[n];
(%o31) s20_n := s01_n + s03_n
(%i32) makelist(s20[n], n, 1, 5);
```

(%o32) $[\sin(1)+\frac{1}{5},\sin(2)+\frac{3}{8},\sin(3)+\frac{5}{11},\sin(4)+\frac{1}{2},\sin(5)+\frac{9}{17}]$

```
(%i33) s21[n]:=s01[n]-s03[n];
(%o33) s21_n := s01_n - s03_n
(%i34) makelist(s21[n], n, 1, 5);
```

(%o34) $[\frac{1}{5}-\sin(1),\frac{3}{8}-\sin(2),\frac{5}{11}-\sin(3),\frac{1}{2}-\sin(4),\frac{9}{17}-\sin(5)]$

```
(%i35) s22[n]:=2*s01[n]+3*n*s02[n];
(%o35) s22_n := 2 s01_n + 3 n s02_n
(%i36) makelist(s22[n], n, 1, 5);
```

(%o36) $[3\sin(1)+\frac{2}{5},6\sin\left(\frac{1}{2}\right)+\frac{3}{4},9\sin\left(\frac{1}{3}\right)+\frac{10}{11},12\sin\left(\frac{1}{4}\right)+1,15\sin\left(\frac{1}{5}\right)+\frac{18}{17}]$

```
(%i37) s23[n]:=s02[n]*s03[n];
(%o37) s23_n := s02_n s03_n
```

(%i38) makelist(s23[n], n, 1, 5);

(%o38) $[\sin(1)^2, \sin\left(\frac{1}{2}\right)\sin(2), \sin\left(\frac{1}{3}\right)\sin(3), \sin\left(\frac{1}{4}\right)\sin(4), \sin\left(\frac{1}{5}\right)\sin(5)]$

(%i39) s24[n]:=s05[n]/s06[n];

(%o39) $s24_n := \frac{s05_n}{s06_n}$

(%i40) makelist(s24[n], n, 1, 5);

(%o40) $[-1, \frac{1}{8}, -\frac{1}{27}, \frac{1}{64}, -\frac{1}{125}]$

Con wxMaxima se puede obtener una representación gráfica de los puntos $(n, f(n))$ de una sucesión, aspecto que puede ayudar a su interpretación, al menos, en una primera aproximación. La sintaxis es la que se puede ver a continuación y hay que observar que se han de definir dos listas, una primera para los valores sucesivos de los números naturales (abscisas) y una segunda para los términos correspondientes de la sucesión (ordenadas):

```
(%i41) N[n]:=n$
       List_0:makelist(N[n], n, 1, 40)$
(%i43) List_1:makelist(s01[n], n, 1, 40)$
       wxplot2d([discrete,List_0,List_1], [style, [points, 2,1,1]] , [xlabel, ""] , [ylabel, ""]  )$
```

(%t44)



Si la instrucción que se aplica es, como en este caso, “wxplot2d”, la gráfica se engancha a la salida. En cambio, si la instrucción es “plot2d”, entonces la salida es una gráfica en un formato llamado “gnuplot”, que es un estándar adoptado por wxMaxima; hay que ir a la parte superior de la gráfica para modificar los parámetros de la representación gráfica. Para copiarlo y pegarlo después en un documento, hay que situar el cursor en la parte superior de la gráfica, pulsando con el botón derecho del ratón, ir a “options” y seleccionar “copy to clipboard”. Una vez hecho esto, hay que ir al documento y “pegar” la imagen.

Otro ejemplo. En este caso se trata de una sucesión de tipo “oscilante”:

```
(%i45) List_2:makelist(s03[n], n, 1, 40)$
wxplot2d([discrete,List_0,List_2], [style, [points, 2,2,2]] , [xlabel, ""] , [ylabel, ""] )$
```

(%t46)



Ahora un ejemplo de sucesión divergente hacia $+\infty$:

```
(%i47) List_3:makelist(s06[n], n, 1, 40)$
wxplot2d([discrete,List_0,List_3], [style,[points, 2,3,3]] , [xlabel, ""] , [ylabel, ""] )$
```

(%t48)



Ahora un ejemplo de sucesión divergente hacia $-\infty$:

```
(%i49) List_4:makelist(s07[n], n, 1, 40)$
wxplot2d([discrete,List_0,List_4], [style, [points, 2,4,4]] , [xlabel, ""] , [ylabel, ""] )$
```

(%t50)



Y ahora un ejemplo de sucesión convergente con sucesiones parciales de diferente comportamiento en la monotonía:

```
(%i56) List_5:makelist(s08[n], n, 1, 40)$
(%i51) List_5:makelist(s08[n], n, 1, 40)$
       wxplot2d([discrete,List_0,List_5], [style, [points, 2,3,4]] , [xlabel, ""] , [ylabel, ""] )$
```

(%t52)



En algunos casos, el término general de una sucesión contiene un número de términos que dependen del índice n; por ejemplo, la sucesión (a_n) definida por

$$a_n = \frac{1}{\sqrt{n^2+3}} + \frac{1}{\sqrt{n^2+6}} + \cdots + \frac{1}{\sqrt{n^2+3n}},\ n \geq 1$$

Por lo tanto, el término a_n está definido con la suma de n números; por lo tanto, cada término contiene un número de términos que dependen del índice n. Con wxMaxima esta sucesión se define mediante la sintaxis siguiente:

```
(%i53) a[n]:=sum(1/sqrt(n^2+3*k), k, 1, n);
       a[1];     a[3];     a[5];
```

(%o53) $a_n := \sum_{k=1}^{n} \frac{1}{\sqrt{n^2+3\,k}}$

(%o54) $\frac{1}{2}$

(%o55) $\frac{1}{\sqrt{15}} + \frac{1}{2\sqrt{3}} + \frac{1}{3\sqrt{2}}$

(%o56) $\frac{1}{\sqrt{37}} + \frac{1}{\sqrt{34}} + \frac{1}{\sqrt{31}} + \frac{1}{2\sqrt{10}} + \frac{1}{2\sqrt{7}}$

Observad que el programa escribe el término n-ésimo efectuando operaciones de reordenación y racionalizando algunas de las expresiones que aparecen. Por ejemplo:

(%i57) a[10];

(%o57) $\frac{1}{\sqrt{130}}+\frac{1}{\sqrt{127}}+\frac{1}{\sqrt{118}}+\frac{1}{\sqrt{115}}+\frac{1}{\sqrt{109}}+\frac{1}{\sqrt{106}}+\frac{1}{\sqrt{103}}+\frac{1}{2\sqrt{31}}+\frac{1}{4\sqrt{7}}+\frac{1}{11}$

(%i58) a[20];

(%o58) $\frac{1}{\sqrt{457}}+\frac{1}{\sqrt{454}}+\frac{1}{\sqrt{451}}+\frac{1}{\sqrt{445}}+\frac{1}{\sqrt{442}}+\frac{1}{\sqrt{439}}+\frac{1}{\sqrt{433}}+\frac{1}{\sqrt{430}}+\frac{1}{\sqrt{427}}+\frac{1}{\sqrt{421}}+\frac{1}{\sqrt{418}}+\frac{1}{\sqrt{415}}+\frac{1}{\sqrt{409}}+\frac{1}{\sqrt{406}}+$
$\frac{1}{\sqrt{403}}+\frac{1}{2\sqrt{115}}+\frac{1}{2\sqrt{109}}+\frac{1}{2\sqrt{106}}+\frac{1}{2\sqrt{103}}+\frac{1}{8\sqrt{7}}$

04-2.- Cálculo del límite de una sucesión.

Los cálculos de este apartado se desarrollan en el fichero **Tema_04-2.wxm**.

Comenzaremos recordando la definición de límite de una sucesión de números reales. Se dice que una sucesión $(x_n)_{n\in\mathbb{N}}$ tiene límite el número real $L \in \mathbb{R}$ si cualquiera que sea $\varepsilon > 0$ existe $n_0 \in \mathbb{N}$, que depende de ε, tal que para todo $n > n_0$ se cumple $|x_n - L| < \varepsilon$.

Una sucesión que tiene límite real se dice convergente; una sucesión que no tiene límite real se dice divergente. La notación habitual es $\lim(x_n) = L$ o bien $(x_n) \to L$. Las sucesiones, en relación a si tienen o no límite, se clasifican en:

- convergentes, cuando tienen límite real;
- divergentes a infinito, cuando su límite es $+\infty$ o $-\infty$.
- oscilantes, cuando no se cumple ninguna de las anteriores.

La interpretación de la definición de límite de una sucesión es bastante conocida: todos los términos de la sucesión a partir del n_0-ésimo se encuentran en el intervalo $]L-\varepsilon, L+\varepsilon[$.

Para comprobar que se cumple esta condición e ilustrarlo adecuadamente, lo haremos con un ejemplo. Consideramos la sucesión de números reales $(x_n) = \left(\frac{2n-1}{3n+2}\right)_{n\in\mathbb{N}}$ y comprobamos que es convergente y de límite $L = \frac{2}{3}$. En primer lugar calculamos la diferencia $|x_n - L|$:

(%i1) a[n]:=(2*n-1)/(3*n+2);

(%o1) $a_n := \frac{2\,n-1}{3\,n+2}$

(%i2) a[n]-2/3;

(%o2) $\frac{2\,n-1}{3\,n+2} - \frac{2}{3}$

(%i3) ratsimp(a[n]-2/3);

(%o3) $-\frac{7}{9\,n+6}$

(%i4) Difer1:-(ratsimp(a[n]-2/3));

(%o4) $\frac{7}{9\,n+6}$

Por lo tanto, se ha de cumplir la equivalencia siguiente:

$$\frac{7}{9n+6}<\varepsilon \quad \forall n>n_0 \quad \Leftrightarrow \quad \varepsilon-\frac{7}{9n+6}=\frac{9\varepsilon n+6\varepsilon-7}{9n+6}>0 \quad \forall n>n_0$$

Esta condición equivale a

$$9\varepsilon n+6\varepsilon-7>0 \quad \Leftrightarrow \quad n>\frac{7-6\varepsilon}{9\varepsilon}$$

Ahora se trata de definir las variables y aplicar el procedimiento con wxMaxima; podemos ver unos ejemplos de algunas iteraciones:

```
(%i5) epsilon:0.01;          n0:(7-6*epsilon)/(9*epsilon);
      assume(n>n0);          is(7/(9*n+6)<epsilon);
(%o5) 0.01
(%o6) 77.11111111111111
(%o7) [n>77.11111111111111]
(%o8) true
(%i9) epsilon:0.0001;        n0:(7-6*epsilon)/(9*epsilon);
      assume(n>n0);          is(7/(9*n+6)<epsilon);
(%o9) 1.0 10^-4
(%o10) 7777.11111111111
(%o11) [n>7777.11111111111]
(%o12) true
(%i13) epsilon:0.00000001;   n0:(7-6*epsilon)/(9*epsilon);
      assume(n>n0);          is(7/(9*n+6)<epsilon);
(%o13) 1.0 10^-8
(%o14) 7.777777111111119 10^7
(%o15) [n>7.777777111111119 10^7]
(%o16) true
```

Se puede ir disminuyendo el valor asignado a $\varepsilon>0$ y verificando que se cumple la condición a partir de un número natural que se va incrementando a medida que $\varepsilon>0$ va disminuyendo.

Cuando se tiene definida una sucesión con wxMaxima como una función de variable discreta, una instrucción sencilla permite calcular el límite de esta sucesión. Se trata de escribir:

limit(término general de la sucesión, n, inf)

Veamos algunos ejemplos.

Sucesión $\left(\dfrac{2n-1}{3n+2}\right)_{n\in\mathbb{N}}$:

```
(%i1) a[n]:=(2*n-1)/(3*n+2);
```

(%o1) $a_n := \dfrac{2\,n-1}{3\,n+2}$

```
(%i2) '(lim(a[n])) = limit(a[n], n, inf);
```

(%o2) $\lim(a_n) = \dfrac{2}{3}$

Sucesión $\left(\sin\left(\dfrac{1}{n}\right)\right)_{n\in\mathbb{N}}$:

```
(%i3) b[n]:=sin(1/n);
```

(%o3) $b_n := \sin\left(\dfrac{1}{n}\right)$

```
(%i4) '(lim(b[n])) = limit(b[n], n, inf);
```

(%o4) $\lim(b_n) = 0$

Sucesión $\left(n^2+2n+1\right)_{n\in\mathbb{N}}$:

```
(%i5) c[n]:=n^2+2*n+1;
```

(%o5) $c_n := n^2+2\,n+1$

```
(%i6) '(lim(c[n])) = limit(c[n], n, inf);
```

(%o6) $\lim(c_n) = \infty$

Sucesión $\left(-n^2+2n+1\right)_{n\in\mathbb{N}}$:

```
(%i7) d[n]:=-n^2+2*n+1;
```

(%o7) $d_n := -n^2+2\,n+1$

```
(%i8) '(lim(d[n])) = limit(d[n], n, inf);
```

(%o8) $\lim(d_n) = -\infty$

Sucesión $\left((-1)^n\right)_{n\in\mathbb{N}}$:

```
(%i9) aa[n]:=(-1)^n;
```

(%o9) $aa_n := (-1)^n$

```
(%i10) '(lim(aa[n])) = limit(aa[n], n, inf);
(%o10) lim(aa_n)=ind
```

Obsérvese que en este caso el programa responde "ind" que significa que no se puede calcular el límite, ya que este no existe.

Sucesión $\left(\left(1+\dfrac{1}{n}\right)^{n}\right)_{n\in\mathbb{N}}$:

```
(%i11) e[n]:=(1+1/n)^n;
```

$$(\%o11)\ e_n := \left(1+\frac{1}{n}\right)^{n}$$

```
(%i12) '(lim(e[n])) = limit(e[n], n, inf);
(%o12) lim(e_n)=%e
```

Vale la pena recordar que se cumple la propiedad siguiente: si $(x_n)\to+\infty$, entonces se cumple:

$$\lim\left(\left(1+\frac{1}{x_n}\right)^{x_n}\right)_{n\in\mathbb{N}}=e$$

A modo de comprobación:

```
(%i13) x[n]:=(1+1/n^2)^(n^2);
```

$$(\%o13)\ x_n := \left(1+\frac{1}{n^2}\right)^{n^2}$$

```
(%i14) '(lim(x[n])) = limit(x[n], n, inf);
(%o14) lim(x_n)=%e
(%i15) y[n]:=(1+1/(3*n^2+5*n+4))^(3*n^2+5*n+4);
```

$$(\%o15)\ y_n := \left(1+\frac{1}{3\,n^2+5\,n+4}\right)^{3\,n^2+5\,n+4}$$

```
(%i16) '(lim(y[n])) = limit(y[n], n, inf);
(%o16) lim(y_n)=%e
```

Límite de una sucesión recurrente.

Para calcular el límite de una sucesión definida por recurrencia no se puede aplicar la sintaxis anterior. En efecto, consideremos la sucesión recurrente definida por:

$$\begin{cases} x_1 = 3 \\ x_n = \sqrt{2 + x_{n-1}}, \quad \forall n \geq 2 \end{cases}$$

Si la ponemos en sintaxis de wxMaxima y calculamos el límite, obtenemos la respuesta siguiente:

```
(%i1) x[1]:3;
      x[n]:=sqrt(2+x[n-1]);
(%o1) 3
(%o2) x_n:=sqrt(2+x_(n-1))
(%i3) '(lim(x[n])) = limit(x[n], n, inf);
Maxima encountered a Lisp error:
 Error in PROGN [or a callee]: Bind stack overflow.
 Automatically continuing.
 To enable the Lisp debugger set *debugger-hook* to r
```

En este tipo de sucesiones hace falta aplicar la metodología siguiente: demostrar que es convergente aplicando los criterios de convergencia, habitualmente, monotonía y acotación. Véase el apartado siguiente. Una vez se ha demostrado que la sucesión es convergente, hay que escribir la ecuación que ha de cumplir el límite, definida a partir de la fórmula recurrente de la sucesión; en este caso la ecuación es:

$$\lim(x_n) = \sqrt{2 + \lim(x_{n-1})} \quad \Leftrightarrow \quad L = \sqrt{2 + L}$$

En wxMaxima la sintaxis es la siguiente:

```
(%i4) Eq1:L=sqrt(2+L);
      solve(Eq1, L);
(%o4) L=sqrt(L+2)
(%o5) [L=sqrt(L+2)]
```

Como vemos, el programa no la puede resolver en esta forma; hace falta escribirla de la forma siguiente:

```
(%i6) Eq2:L^2=(2+L);
      solve(Eq2, L);
(%o6) L^2=L+2
(%o7) [L=2,L=-1]
```

Aplicando las propiedades de la sucesión, se puede afirmar que no tiene sentido que el límite sea negativo y, por lo tanto, deducimos que $L = 2$.

Hay que insistir en que esta metodología tan solo es aplicable una vez se ha demostrado que la sucesión es convergente. Por ejemplo, si se considera la sucesión recurrente definida por:

$$\begin{cases} x_1 = 1 \\ x_n = 2x_{n-1} + 1, \ \ \forall n \geq 2 \end{cases}$$

y se quiere calcular el límite aplicando directamente la metodología descrita antes, se puede concluir que $\lim(x_n) = -1$:

```
(%i8) y[1]:1$
      y[n]:=2*y[n-1]+1;
(%o9) y_n:=2 y_(n-1)+1
(%i10) Eq3:L=2*L+1;
       solve(Eq3, L);
(%o10) L=2 L+1
(%o11) [L=-1]
```

Esta conclusión es evidentemente falsa, ya que es inmediato comprobar que $x_n = 2^n - 1, \ n \geq 1$ i, por lo tanto, que se cumple $\lim(x_n) = +\infty$.

En el caso de la sucesión de Fibonacci, definida por:

$$\begin{cases} x_1 = 1, \ \ x_2 = 1 \\ x_n = x_{n-2} + x_{n-1}, \ \ \forall n \geq 3 \end{cases}$$

es evidente que se trata de una sucesión divergente a $+\infty$. En los ejercicios se propone una cuestión interesante en relación a esta sucesión (ver ejercicio 04.10).

04-3.- Formas indeterminadas o indeterminaciones

Los cálculos de este apartado se desarrollan en el fichero **Tema_04-3.wxm**.

Se llaman formas indeterminadas o indeterminaciones las sucesiones construidas con operaciones entre otras sucesiones, de manera que inicialmente no se puede afirmar nada sobre el límite de la sucesión. Veamos unos ejemplos.

Ejemplo 04-3.1.- Consideremos las sucesiones numéricas:

$$x_n = \left(n^2+1\right);\quad y1_n = \left(n^2-1\right);\quad y2_n = \left(n^3-1\right);\quad y3_n = \left(n-1\right)$$

Todas ellas cumplen que tienen límite infinito. Calculemos ahora el límite de la sucesión diferencia de la primera con cada una de las otras:

```
(%i1) x[n]:=n^2+1;   y1[n]:=n^2-1;   y2[n]:=n^3-1;   y3[n]:=n-1;
```

(%o1) $x_n := n^2 + 1$

(%o2) $y1_n := n^2 - 1$

(%o3) $y2_n := n^3 - 1$

(%o4) $y3_n := n - 1$

```
(%i5) L1=limit(x[n]-y1[n], n, inf);
      L2=limit(x[n]-y2[n], n, inf);
      L3=limit(x[n]-y3[n], n, inf);
```

(%o5) $L1 = 2$

(%o6) $L2 = -\infty$

(%o7) $L3 = \infty$

Como se ve, en cada caso da un resultado diferente, de manera que de entrada no se puede afirmar nada sobre cuál será el límite de la diferencia de dos sucesiones de límite infinito y, por lo tanto, esta diferencia es una forma indeterminada.

Ejemplo 04-3.2.- Consideremos les sucesiones numéricas:

$$x_n = \left(\frac{1}{n^2+1}\right);\quad y1_n = \left(n^2-1\right);\quad y2_n = \left(n^3-1\right);\quad y3_n = \left(n-1\right)$$

La primera es un infinitésimo y las otras tres son infinitos. Calculemos ahora el límite de la sucesión producto de la primera con cada una de las otras:

```
(%i1) x[n]:=1/(n^2+1);    y1[n]:=n^2-1;
      y2[n]:=n^3-1;       y3[n]:=n-1;
```

(%o1) $x_n := \frac{1}{n^2+1}$

(%o2) $y1_n := n^2 - 1$

(%o3) $y2_n := n^3 - 1$

(%o4) $y3_n := n - 1$

```
(%i5) L1=limit(x[n]*y1[n], n, inf);
      L2=limit(x[n]*y2[n], n, inf);
      L3=limit(x[n]*y3[n], n, inf);
```

(%o5) $L1 = 1$

(%o6) $L2 = \infty$

(%o7) $L3 = 0$

Como se ve, en cada caso da un resultado diferente, de manera que de entrada no se puede afirmar nada sobre cuál será el límite del producto de dos sucesiones siendo la primera un infinitésimo y la segunda un infinito y, por lo tanto, este producto es una forma indeterminada. En la Tabla 04-1 hay las formas indeterminadas o indeterminaciones para sucesiones numéricas y la metodología de resolución, caso de que no se pueda obtener el límite con wxMaxima.

Tipos de indeterminación	**Sucesiones**	**Forma indeterminada**	**Metodología de resolución**
$\infty - \infty$	$(x_n) \to +\infty;\ (y_n) \to +\infty$	$\lim(x_n - y_n)$	Multiplicar y dividir por la suma $(x_n + y_n)$
$0 . \infty$	$(a_n) \to 0;\ (x_n) \to +\infty$	$\lim(a_n x_n)$	Criterio de Stolz
$0/0$	$(a_n) \to 0;\ (b_n) \to 0$	$\lim(a_n / b_n)$	Criterio de Stolz
∞/∞	$(x_n) \to +\infty;\ (y_n) \to +\infty$	$\lim(x_n / y_n)$	Criterio de Stolz
1^∞	$(\lambda_n) \to 1;\ (x_n) \to +\infty$	$\lim(x_n^{\lambda_n})$	Reducción al número *e*
0^0	$(a_n) \to 0;\ (b_n) \to 0$	$\lim(a_n^{b_n})$	Aplicar logaritmos y después el criterio de Stolz
∞^0	$(a_n) \to 0;\ (x_n) \to +\infty$	$\lim(x_n^{a_n})$	Aplicar logaritmos y después el criterio de Stolz

Tabla 04-1. Formas indeterminadas para sucesiones numéricas

04-4.- Criterios de convergencia y cálculo de límites de sucesiones numéricas.

Los cálculos de este apartado se desarrollan en el fichero **Tema_04-4.wxm**.

En este apartado se ven algunos criterios de convergencia y métodos de cálculo de límites que se aplican más a menudo en el análisis de la convergencia y en el cálculo de límites de sucesiones.

Ejemplo 04-4.1 (Criterio de compresión). Se considera la sucesión $(a_n)_{n\in\mathbb{N}}$ de números reales definida por:

$$a_n = \frac{1}{\sqrt{n^2+3}} + \frac{1}{\sqrt{n^2+6}} + \cdots + \frac{1}{\sqrt{n^2+3n}},\ n \geq 1$$

Si la introducimos en el programa wxMaxima y queremos calcular el límite obtenemos:

```
(%i1) a[n]:=sum(1/sqrt(n^2+3*k), k, 1, n);
```

(%o1) $a_n := \sum_{k=1}^{n} \frac{1}{\sqrt{n^2+3\,k}}$

```
(%i2) '(lim(a[n])) = limit(a[n], n, inf);
```

(%o2) $\mathrm{lim}(a_n) = \lim_{n\to\infty} \sum_{k=1}^{n} \frac{1}{\sqrt{n^2+3\,k}}$

Por lo tanto, no es posible hacer el cálculo de este límite de la manera sencilla habitual. Entonces observamos lo siguiente: por un lado se cumple:

$$a_n = \frac{1}{\sqrt{n^2+3}} + \frac{1}{\sqrt{n^2+6}} + \cdots + \frac{1}{\sqrt{n^2+3n}} \leq \frac{1}{\sqrt{n^2+3}} + \cdots + \frac{1}{\sqrt{n^2+3}} = \frac{n}{\sqrt{n^2+3}} = y_n$$

Por otro lado:

$$a_n = \frac{1}{\sqrt{n^2+3}} + \frac{1}{\sqrt{n^2+6}} + \cdots + \frac{1}{\sqrt{n^2+3n}} \geq \frac{1}{\sqrt{n^2+3n}} + \cdots + \frac{1}{\sqrt{n^2+3n}} = \frac{n}{\sqrt{n^2+3n}} = x_n$$

Ahora podemos calcular los límites de estas sucesiones con wxMaxima:

```
(%i3) x[n]:=n/sqrt(n^2)$
(%i4) '(lim(x[n])) = limit(x[n], n, inf);
```
(%o4) $\lim(x_n)=1$

```
(%i5) y[n]:=n/sqrt(n^2+3*n)$
(%i6) '(lim(y[n])) = limit(y[n], n, inf);
```
(%o6) $\lim(y_n)=1$

Por lo tanto, por aplicación del criterio de compresión, se cumple $\lim(a_n)=1$.

Ejemplo 04-4.2 (Teorema de la convergencia monótona). Se considera la sucesión $(a_n)_{n\in\mathbb{N}}$ de números reales definida por:

$$\begin{cases} a_1 = 3 \\ a_n = \sqrt{2+a_{n-1}}, \; n \geq 2 \end{cases}$$

Como antes, wxMaxima no da el límite de la sucesión. En efecto:

```
(%i1) a[1]:3;        a[n]:=sqrt(2+a[n-1]);
```
(%o1) 3

(%o2) $a_n := \sqrt{2+a_{n-1}}$

```
(%i3)  makelist(a[n], n, 1, 5);
```
(%o3) $[3,\sqrt{5},\sqrt{\sqrt{5}+2},\sqrt{\sqrt{\sqrt{5}+2}+2},\sqrt{\sqrt{\sqrt{\sqrt{5}+2}+2}+2}]$

```
(%i4) '(lim(a[n])) = limit(a[n], n, inf);
Maxima encountered a Lisp error:
 Error in PROGN [or a callee]: Bind stack overflow.
 Automatically continuing.
 To enable the Lisp debugger set *debugger-hook* to r
```

Calculemos ahora los primeros términos de esta sucesión:

```
(%i5) a[1];     a[2],numer;     a[3],numer;
      a[4],numer;     a[5],numer;
(%o5) 3
 (%o6) 2.23606797749979
 (%o7) 2.058171027271492
 (%o8) 2.014490264873845
 (%o9) 2.0036192914009
```

Esto permite intuir que la sucesión es monótona decreciente; en efecto, si se supone que $a_n > a_{n-1}$ (hipótesis de inducción), entonces se cumple $2+a_n > 2+a_{n-1}$ y por lo tanto $\sqrt{2+a_n} > \sqrt{2+a_{n-1}}$, es decir, $a_{n+1} > a_n$. Por lo tanto, la sucesión $(a_n)_{n\in\mathbb{N}}$ es monótona decreciente. Para ver que es acotada inferiormente, es suficiente observar que por definición de la sucesión se cumple que $a_n > 0, \forall n \geq 1$. Por lo tanto, la sucesión es convergente, en virtud del teorema de la convergencia monótona. Ahora ya se puede aplicar la metodología de cálculo del límite:

```
(%i10) Eq32:L=sqrt(2+L);
```
$$(\%o10)\ L=\sqrt{L+2}$$
```
(%i11) Eq32a:L^2=L+2;
       solve(Eq32a, L);
```
$$(\%o11)\ L^2=L+2$$
$$(\%o12)\ [L=2, L=-1]$$

En definitiva, se puede afirmar que $\lim(a_n)=2$.

Ejemplo 04-4.3 (Criterio de la raíz *n*-ésima). Se considera la sucesión $(a_n)_{n\in\mathbb{N}}$ de números reales definida por:

$$a_n = \left(2n^2+1\right)^{1/n}, \; n \geq 1$$

Definimos la sucesión con wxMaxima y calculamos los primeros términos de esta sucesión:

```
(%i1) a[n]:=(2*n^2+1)^(1/n);
```
$$(\%o1)\ a_n:=(2\,n^2+1)^{1/n}$$
```
(%i2) [a[1], a[2],a[3],a[4],a[5]];
```
$$(\%o2)\ [3,3,19^{1/3},33^{1/4},51^{1/5}]$$
```
(%i3) [a[1], a[2],a[3],a[4],a[5]], numer;
(%o3) [3,3,2.668401648721944,2.39678172692843,2.19540189742749]
```

Si construimos la sucesión $(b_n)_{n\in\mathbb{N}}$ con $b_n = \dfrac{a_{n+1}}{a_n}$, se obtendrá:

```
(%i4) b[n]:=a[n+1]/a[n];
```
$$(\%o4)\ b_n:=\frac{a_{n+1}}{a_n}$$

```
(%i5) makelist(b[n], n, 1, 5);
```

$$(\%o5)\ [1,\frac{19^{1/3}}{3},\frac{33^{1/4}}{19^{1/3}},\frac{51^{1/5}}{33^{1/4}},\frac{73^{1/6}}{51^{1/5}}]$$

```
(%i6) makelist(float(b[n]), n, 1, 5);
(%o6) [1.0,0.88946721624065,0.89820875657021,0.91597907008453,0.93119315475983]
```

Calculamos ahora el límite de esta sucesión:

```
(%i7) '(lim(b[n])) = limit(b[n], n, inf);
```

$$(\%o7)\ \lim(b_n)=1$$

Por lo tanto, se cumple $\lim(a_n)=1$.

Ejemplo 04-4.4 (Criterio de Stolz). Se considera la sucesión $(a_n)_{n\in\mathbb{N}}$ de números reales definida por:

$$a_n=\frac{1+\frac{1}{2}+\cdots+\frac{1}{n}}{\log(n)},\ n\geq 2$$

Definimos en primer lugar la sucesión del numerador:

```
(%i1) x[k]:=1/k$        X[n]:=sum(x[k], k, 1, n);
```

$$(\%o2)\ X_n:=\sum_{k=1}^{n}x_k$$

```
(%i3) makelist(X[n], n, 1, 10);
```

$$(\%o3)\ [1,\frac{3}{2},\frac{11}{6},\frac{25}{12},\frac{137}{60},\frac{49}{20},\frac{363}{140},\frac{761}{280},\frac{7129}{2520},\frac{7381}{2520}]$$

Ahora la sucesión del denominador y la sucesión cociente:

```
(%i4) Y[n]:=log(n);
      a[n]:=X[n]/Y[n];
```

$$(\%o4)\ Y_n:=\log(n)$$

$$(\%o5)\ a_n:=\frac{X_n}{Y_n}$$

Calculamos el límite de esta sucesión:

(%i6) '(lim(a[n])) = limit(a[n], n, inf);

$$\text{(\%o6)}\ \lim(a_n) = \lim_{n \to \infty} \frac{\sum_{k=1}^{n} \frac{1}{k}}{\log(n)}$$

Esta respuesta del programa sugiere que no se puede calcular el límite de la manera estándar. Aplicaremos el criterio de Stolz, una vez se haya verificado que se cumplen las condiciones para su aplicación:

(%i7) xx[n]:=x[n+1]; YY[n]:=Y[n+1]-Y[n]; aa[n]:=xx[n]/YY[n]; makelist(aa[k], k, 1, 4);

$$\text{(\%o7)}\ xx_n := x_{n+1}$$

$$\text{(\%o8)}\ YY_n := Y_{n+1} - Y_n$$

$$\text{(\%o9)}\ aa_n := \frac{xx_n}{YY_n}$$

$$\text{(\%o10)}\ \left[\frac{1}{2\log(2)}, \frac{1}{3(\log(3)-\log(2))}, \frac{1}{4(\log(4)-\log(3))}, \frac{1}{5(\log(5)-\log(4))}\right]$$

Ahora calculamos el límite de esta sucesión:

(%i11) '(lim(aa[n])) = limit(aa[n], n, inf);

$$\text{(\%o11)}\ \lim(aa_n) = 1$$

Por lo tanto, se cumple $\lim(a_n) = 1$.

04-5.- Equivalencia de sucesiones

Los cálculos de este apartado se desarrollan en el fichero **Tema_04-5.wxm**.

El concepto de equivalencia de sucesiones y su aplicación al cálculo de límites es puede ver en las referencias, particularmente en [BU07] y [JB01]. En este apartado se muestran algunos ejemplos de sucesiones equivalentes y se ilustra la aplicación de este concepto al cálculo de límites de sucesiones.

Recordemos que dos sucesiones numéricas $(a_n),(b_n)$ se dice que son equivalentes, si, y sólo si, se cumple $\lim \frac{a_n}{b_n} = 1$. La notación habitual para designar este hecho es $(a_n) \sim (b_n)$. Por ejemplo, las sucesiones numéricas $a_n = \frac{2n}{3n+1}$ i $b_n = \frac{2n+4}{3n+7}$ son equivalentes. Si dos sucesiones son convergentes y equivalentes, entonces tienen el mismo límite; lo recíproco no es cierto, es decir, dos sucesiones pueden ser convergentes y del mismo límite pero pueden no ser equivalentes, como por ejemplo las sucesiones $a_n = \frac{2n}{3n^2+1}$ i $b_n = \frac{2n+4}{3n^4+7}$.

La propiedad más interesante a efectos prácticos de las sucesiones equivalentes es la siguiente: si $(a_n) \sim (\alpha_n)$ y $L = \lim(a_n b_n)$ existe, entonces $L = \lim(\alpha_n b_n)$. Es decir, en productos y cocientes se pueden sustituir algunos factores por factores equivalentes.

Veamos a continuación algunos ejemplos de sucesiones equivalentes.

Ejemplo 04-5.1. Si (x_n) es un infinitésimo, es decir es tal que $(x_n) \to 0$, entonces se cumple:

$$(x_n) \sim (\sin(x_n)) \quad ; \quad (x_n) \sim (\tan(x_n))$$

Veamos, por ejemplo, el caso en que $x_n = \frac{1}{n^2+1}$:

```
(%i1) x[n]:=1/(n^2+1)$      y[n]:=sin(x[n])$      a[n]:=x[n]/(y[n])$
      'lim(a[n])=limit(a[n], n, inf);
```

(%o4) $\lim\left(\frac{1}{(n^2+1)\sin\left(\frac{1}{n^2+1}\right)}\right) = 1$

```
(%i5) z[n]:=tan(x[n])$        b[n]:=x[n]/(z[n])$
      'lim(b[n])=limit(b[n], n, inf);
```

(%o7) $\lim\left(\frac{1}{(n^2+1)\tan\left(\frac{1}{n^2+1}\right)}\right) = 1$

Ejemplo 04-5.2. Si $(x_n) \to 0$ entonces se cumple:

$$(x_n^2/2) \sim (1-\cos(x_n))$$

Veamos un ejemplo:

```
(%i1) x[n]:=1/(n^2+1)$     y[n]:=1-cos(x[n])$
      a[n]:=(y[n])/(x[n]^2/2);
```

(%o3) $a_n := \dfrac{y_n}{\frac{x_n^2}{2}}$

```
(%i4) 'lim(a[n])=limit(a[n], n, inf);
```

(%o4) $\lim\left(2\,(n^2+1)^2\left(1-\cos\left(\dfrac{1}{n^2+1}\right)\right)\right)=1$

Ejemplo 04-5.3. Si $(x_n) \to 0$ entonces se cumple:

$$(x_n) \sim (e^{x_n}-1)\,; \quad (1+x_n) \sim (e^{x_n})$$

Veamos un caso concreto:

```
(%i1) x[n]:=1/(n^2+1)$     y[n]:=exp(x[n])-1$     a[n]:=(y[n])/x[n]$
      'lim(a[n])=limit(a[n], n, inf);
```

(%o4) $\lim\left((n^2+1)\left(\%e^{\frac{1}{n^2+1}}-1\right)\right)=1$

```
(%i5) xx[n]:=1+x[n]$     yy[n]:=exp(x[n])$     aa[n]:=yy[n]/xx[n]$
      'lim(aa[n])=limit(aa[n], n, inf);
```

(%o8) $\lim\left(\dfrac{\%e^{\frac{1}{n^2+1}}}{\frac{1}{n^2+1}+1}\right)=1$

Ejemplo 04-5.4. Si $(x_n) \to 0$ entonces se cumple:

$$(x_n) \sim \left(\log(x_n+1)\right)$$

Veamos un caso concreto:

```
(%i1) x[n]:=1/(n^2+1)$        y[n]:=log(x[n]+1)$
(%i3) a[n]:=y[n]/x[n];
      'lim(a[n])=limit(a[n], n, inf);
```

(%o3) $a_n := \frac{y_n}{x_n}$

(%o4) $\lim\left((n^2+1)\log\left(\frac{1}{n^2+1}+1\right)\right)=1$

Ejemplo 04-5.5. Si $(y_n)\to 1$ entonces se cumple:

$$(y_n-1)\sim\left(\log(y_n)\right)$$

Veamos un caso concreto:

```
(%i1) x[n]:=log(y[n]);    y[n]:=n^2/(n^2+1);
```

(%o1) $x_n := \log(y_n)$

(%o2) $y_n := \frac{n^2}{n^2+1}$

```
(%i3) z[n]:=y[n]-1;       a[n]:=x[n]/z[n];
      'lim(a[n])=limit(a[n], n, inf);
```

(%o3) $z_n := y_n - 1$

(%o4) $a_n := \frac{x_n}{z_n}$

(%o5) $\lim\left(\frac{\log\left(\frac{n^2}{n^2+1}\right)}{\frac{n^2}{n^2+1}-1}\right)=1$

Ejemplo 04-5.6. Para calcular el límite

$$\lim\frac{3n\left(1+\frac{1}{2}+\frac{1}{3}+\cdots+\frac{1}{n}\right)}{(n+2)\log\left(\frac{1}{n}\right)\cos\left(\frac{\pi n}{4n+1}\right)}$$

se implementan los cálculos con wxMaxima y obtenemos:

(%i1) x[n]:=(3*n)*sum(1/k, k, 1, n);

$$\text{(\%o1) } x_n := 3\, n \sum_{k=1}^{n} \frac{1}{k}$$

(%i2) y[n]:=(n+2)*(log(1/n))*(cos((%pi*n)/(4*n+1)));

$$\text{(\%o2) } y_n := (n+2)\log\left(\frac{1}{n}\right)\cos\left(\frac{\pi\, n}{4\, n+1}\right)$$

(%i3) a[n]:=x[n]/y[n]$ '(lim(a[n])) = limit(a[n], n, inf);

$$\text{(\%o4) } \lim(a_n) = 3\left(\lim_{n\to\infty} \frac{\left(\sum_{k=1}^{n}\frac{1}{k}\right) n}{\log\left(\frac{1}{n}\right)(n+2)\cos\left(\frac{\pi\, n}{4\, n+1}\right)}\right)$$

Aplicando la metodología indicada para sucesiones equivalentes y teniendo en cuenta el Ejemplo 04-4.4, el límite anterior es el mismo que

$$\lim \frac{3n}{(n+2)\cos\left(\frac{\pi n}{4n+1}\right)}$$

Con wxMaxima se obtiene el resultado:

(%i5) b[n]:=(3*n)/((n+2)*(cos((%pi*n)/(4*n+1))));
limit(b[n], n, inf);

$$\text{(\%o5) } b_n := \frac{3\, n}{(n+2)\cos\left(\frac{\pi\, n}{4\, n+1}\right)}$$

$$\text{(\%o6) } 3\sqrt{2}$$

Se pueden aplicar las propiedades de las sucesiones equivalentes en sucesiones numéricas divergentes a ∞. Si la sucesión numérica es polinómica:

$$\left(a_p n^p + a_{p-1} n^{p-1} + \cdots + a_1 n + a_0\right)$$

la equivalencia más importante y de aplicación habitual es

$$\left(a_p n^p + a_{p-1} n^{p-1} + \cdots + a_1 n + a_0\right) \sim \left(a_p n^p\right)$$

es decir, la aplicación de la propiedad:

$$\lim \frac{a_p n^p + a_{p-1} n^{p-1} + \cdots + a_1 n + a_0}{a_p n^p} = 1$$

Finalmente, una equivalencia muy útil es la de Stirling: $(n!) \sim \left(n^n e^{-n} \sqrt{2\pi n}\right)$. En efecto:

```
(%i1) a[n]:=n!;       b[n]:=(n^n)*(exp(-n))*sqrt(2*%pi*n);
```

(%o1) $a_n := n!$

(%o2) $b_n := n^n \exp(-n) \sqrt{2 \pi n}$

```
(%i3) s[n]:=a[n]/b[n];
      '(lim(s[n])) = limit(s[n], n, inf);
```

(%o3) $s_n := \frac{a_n}{b_n}$

(%o4) $\lim(s_n) = 1$